⊠응용논문

다구찌의 파라미터 설계에 대한 반응표면 접근방법을 이용한 다반응 최적화*

이우선 · 이종협
성신여자대학교 통계학과
임성수
고려대학교 정보통계학과

Multiresponse Optimization Using a Response Surface Approach to Taguchi's Parameter Design

Woo Sun Lee · Jong Hyup Lee
Dept. of Statistics, Sungshin Women's University
Sungsue Rheem
Dept. of Informational Statistics, Korea University

Abstract

Taguchi's parameter design seeks proper choice of levels of controllable factors (*parameters* in Taguchi's terminology) that makes the quality characteristic of a product optimal while making its variability small. This aim can be achieved by response surface techniques that allow flexibility in modeling and analysis. In this article, a collection of response surface modeling and analysis techniques is proposed to deal with the multiresponse optimization problem in experimentation with Taguchi's signal and noise factors.

* 이 논문은 1997년 성신여자대학교 연구소 특별 사업계획 지원에 따른 기초과학연구소 특별 과제인 "품질향상을 위한 통계적 방법 -다구찌 방법의 개선과 적용방법을 중심으로-"에 대한 연구결과임.

1. 서론

1.1 문제 제기

제어할 수 없는 잡음적 환경(uncontrollable noisy environment) 속에서도 품질특성치가 변동이 작으면서 최적의 값을 가지도록 제어가능한 요인(controllable factor, 이후 간단히 제어요인(control factor)으로 표기함, 다구찌는 이를 신호요인(signal factor)으로 표기하였음)들의 수준값들을 정하는 문제가 파라미터 설계(parameter design) 문제이다. 이 문제는 다구찌(田口, Genichi Taguchi)에 의하여 제기되었고, 다구찌는 이 문제의 해결책으로 제어요인 수준의 내측배열(inner array)과 잡음요인 수준의 외측배열(outer array)를 교차시킨 배열(crossed array)에 의한 실험설계방법과 SN비(signal-to-noise ratio)에 의한 실험자료분석방법을 제안하였다. 그러나, 다구찌의 철학은 환영을 받은 반면에, 다구찌의 방법론은 여러 통계학자들에 의하여, 실험의 규모가 너무 크면서도 제어요인들간의 상호작용(interaction)들을 평가하지 못하는 등 비경제적이고 비효율적이며, 분석방법도 애매하다는 비판을 받아 왔다. 다구찌 방법을 비판한 통계학자들은 대안들을 제시하였는데, 그 중 대표적인 것이 반응표면방법론적 대안으로 실험설계에서는 제어요인의 수준배열과 잡음요인의 수준배열을 결합시킨 배열(combined array)을 사용하고 실험분석에서는 평균과 분산의 동시모형화(simultaneous modeling of mean and variance)를 이용하는 방법이다(Vining and Myers, 1990; Myers, Khuri and Vining, 1992; Myers and Montgomery, 1995).

실제 실험연구에서는 연구자가 관심을 갖는 품질특성치가 둘 이상인 경우가 종종 있다. 파라미터 설계가 아닌 일반적인 반응표면설계에 의한 실험에서 여러 품질특성치들의 평균들을 동시적으로 최적화하는 문제는 Myers and Carter(1973, dual response approach), Derringer and Suich(1980, desirability function approach), Khuri and Conlon(1981, generalized distance approach) 등에 의하여 다루어져 왔다. 본 연구에서는 양적(quantitative) 품질특성치들이 둘 이상 있을 때에 품질특성치들의 평균들을 동시적으로 최적화하며 품질특성치들의 잡음요인에 의한 변동들을 동시적으로 최소화하는 제어요인들의 수준값들을 찾는 문제를 제기하고, 이 문제를 정교하고 융통성있게 해결하는 방법을 제시하고자 한다.

1.2 실험 설계

실험설계로는 다구찌식 교차배열과 반응표면적 결합배열이 모두 사용가능한데, 반응표면적 결합배열이 보다 경제적이며 효율적이라고 본다. 제어요인이 셋(x1, x2, x3)이고 잡음요인이 둘(z1, z2)인 경우에 쓰일 수 있는 실험설계인 다구찌식 교차배열과 중심합성설계(central composit design, 이후 CCD로 표기함)에 근거한 반응표면적 결합배열은 각각 <표 1>, <표 2>와 같다.

< 표 1 > 다구찌식 교차배열 ($L_9 \times 2^2$)

시행번호	x1	x2	x3	z1	z2
1	-1	-1	-1	-1	-1
2	-1	-1	-1	-1	1
3	-1	-1	-1	1	-1
4	-1	-1	-1	1	1
5	-1	0	0	-1	-1
6	-1	0	0	-1	1
7	-1	0	0	1	-1
8	-1	0	0	1	1
9	-1	1	1	-1	-1
10	-1	1	1	-1	1
11	-1	1	1	1	-1
12	-1	1	1	1	1
13	0	-1	0	-1	-1
14	0	-1	0	-1	1
15	0	-1	0	1	-1
16	0	-1	0	1	1
17	0	0	1	-1	-1
18	0	0	1	-1	1
19	0	0	1	1	-1
20	0	0	1	1	1
21	0	1	-1	-1	-1
22	0	1	-1	-1	1
23	0	1	-1	1	-1
24	0	1	-1	1	1
25	1	-1	1	-1	-1
26	1	-1	1	-1	1
27	1	-1	1	1	-1
28	1	-1	1	1	1
29	1	0	-1	-1	-1
30	1	0	-1	-1	1
31	1	0	-1	1	-1
32	1	0	-1	1	1
33	1	1	0	-1	-1
34	1	1	0	-1	1
35	1	1	0	1	-1
36	1	1	0	1	1

< 표 2 > 반응표면적 결합배열

시행번호	x1	x2	x3	z1	z2
1	-1	-1	-1	-1	1
2	-1	-1	-1	1	-1
3	-1	-1	1	-1	-1
4	-1	-1	1	1	1
5	-1	1	-1	-1	-1
6	-1	1	-1	1	1
7	-1	1	1	-1	1
8	-1	1	1	1	-1
9	1	-1	-1	-1	-1
10	1	-1	-1	1	1
11	1	-1	1	-1	1
12	1	-1	1	1	-1
13	1	1	-1	-1	1
14	1	1	-1	1	-1
15	1	1	1	-1	-1
16	1	1	1	1	1
17	-2	0	0	0	0
18	2	0	0	0	0
19	0	-2	0	0	0
20	0	2	0	0	0
21	0	0	-2	0	0
22	0	0	2	0	0
23	0	0	0	0	0
24	0	0	0	0	0
25	0	0	0	0	0
26	0	0	0	0	0

<표 1>의 다구찌식 교차배열은 시행회수가 36이나 되지만 제어요인(x1, x2, x3)과 잡음요인(z1, z2) 사이의 상호작용(interaction)들만 평가할 수 있을 뿐—품질특성치의 변동이라는 비균일성은 제어요인과 잡음요인 사이의 상호작용으로 볼 수 있음—제어요인들간의 상호작용은 평가할 수 없다. 이에 비해 중심에서의 시행(center run)회수가 4인 <표 2>의 결합배열은 시행회수가 26밖에 되지 않으면서도, 시행번호 1부터 16까지의 부분(2^{5-1} 부분설계)이 Resolution V인 것에 기인하여, 제어요인과 잡음요인 사이의 상호작용들뿐만 아니라 제어요인들간의 상호작용도 평가할 수 있는 이점(利點)을 가지고 있다. 중심에서의 시행회수를 n_0이라 할 때, <표 2>에서와 같은 결합배열에서의 시행회수는 22+ n_0 로 <표 1>에서의 시행회수 36에 비해 경제적임을 알

수 있다.

다구찌 방법을 옹호하는 사람들은 <표 1>의 교차배열이, 각 제어요인의 수준조합(여기서는 9종류)이 2^2 설계에 의한 잡음요인의 모든 수준들과 만나게 되어 있어, 잡음요인들의 변화에 가장 둔감한(robust) 제어요인 수준조합을 찾기에 보다 적절하다고 주장할 수 있다. 이 주장은 물론 맞는 말이다. 그러나, <표 2>의 결합배열에 의해서도 제어요인과 잡음요인간 상호작용효과의 추정과 이에 근거한 분산의 모형화를 통하여 잡음요인들의 변화에 둔감한 제어요인 수준들을 추정할 수 있다. 그리고, <표 1>의 교차배열에서는 제어요인 수준조합들은 9종류(-1,-1,-1; -1,0,0; -1,1,1; 0,-1,0; 0,0,1; 0,1,-1; 1,-1,1; 1,0,-1; 1,1,0)이고 잡음요인 수준조합들은 4종류(-1,-1; -1,1; 1,-1; 1,1)이지만, <표 2>의 결합배열에서는 제어요인 수준조합들은 15종류(-1,-1,-1; -1,-1,1; -1,1,-1; -1,1,1; 1,-1,-1; 1,-1,1; 1,1,-1; 1,1,1; -2,0,0; 2,0,0; 0,-2,0; 0,2,0; 0,0,-2; 0,0,2; 0,0,0)이고 잡음요인 수준조합들은 5종류(-1,-1; -1,1; 1,-1; 1,1; 0,0)로, 반응표면적 결합배열이 보다 다양한 조건들에서의 시행결과를 고려한 모형을 제공할 수 있음을 알 수 있다. 또한, 실험의 목적이 잡음요인들의 변화에 가장 둔감한 제어요인 수준조합을 찾는 것만은 아니다. 대부분의 경우, 분산을 줄이면서 평균값을 적절하게 하는 타협(compromise)이 필요하고, 공정에 대한 보다 정확한 이해도 필요하다. <표 2>와 같은 결합배열은, 제어요인들간의 상호작용까지 평가가능하게 하고 평균과 분산의 개별적 모형화까지 가능하게 하는, 모형화에 있어서의 융통성(flexibility in modeling)을 제공하고, 이와 같은 모형화는 공정의 변동성(process variability)과 같은 공정현상을 이해하는 데 도움을 준다. 이와 같이 경제성과 모형화에서의 융통성이라는 두 마리 토끼를 잡게 해 주는 반응표면적 결합배열이 다구찌식 교차배열보다 더 우수하다고 할 수 있다(임성수, 1995; Myers, Khuri and Vining, 1992).

1.3 실험 분석

실험분석은 다음의 4 단계로 진행된다.

(i) 각 품질특성치에 대하여 제어요인들과 잡음요인들로 표현되는 반응표면모형(기본모형)을 잘 적합시키기 위해 제어요인들과 잡음요인들로 이루어지는 모형 항(model term)들의 후보군에서 모형에 들어갈 항들을 회귀분석에서의 변수선택기법을 이용하여 골라낸다. 먼저 2차 다항 모형(second-order polynomial model)을 적합시킨 후, 적합결여 검정(test of lack of fit)을 실시하여 적합결여가 유의하지 않으면 2차 모형 항들을 대상으로 변수선택을 실시하고, 2차 모형의 적합결여가 유의하면 3차 모형 항들을 대상으로 변수선택을 실시한다.

(ii) 각 품질특성치에 대한 기본모형에 들어갈 항들이 모두 결정되면, 여러 품질특성치들에 대한 모형들을 동시에 추정한다. 일반적으로 같은 실험조건 하에서 측정되는 품질특성치들간에 상관(correlation)이 존재할 수 있고 각 품질특성치에 대한 기본모형에 들어갈 항들이 모두 같지는 않은 것이 보통의 경우이므로, SUR(seemingly unrelated regression) 모형을 전체 관측치들에 대한 모형으로 사용하여 모형의 회귀계수들을 FGLS(feasible generalized least squares) 방법으로 추정한다.

(iii) 각각의 기본모형으로부터 평균과 분산의 동시모형화를 실시한다. 평균모형과 분산모형은 제어요인들로만 표현된다. 품질특성치가 k개 있다면 k개의 평균모형과 k개의 분산모형이 구해진다.

(iv) 실험구역 내의 격자(grid)점들에서 k개의 평균들에 대하여 바람직성(desirability) 함수값들을 구하고 이를 결합하여 평균들의 바람직성(desirability of means) 함수값을 구한다. 또한, k개의 표준편차들에 대하여도 바람직성 함수값들을 구하고 이를 결합하여 표준편차들의 바람직성(desirability of standard deviations) 함수값을 구한다. 다음, 평균 바람직성 함수와 표준편차 바람직성 함수를 적절한 가중치를 주어 결합하여 평균들과 표준편차들의 종합 바람직성(overall desirability of means and standard deviations) 함수를 정의하고, 이 종합 바람직성 함수값을 최대화하는 제어요인들의 수준값들을 찾는다.

제 2절에서는 위의 (i), (ii), (iii), (iv)를 구체적으로 설명한다. 실험데이타를 이용한 예시도 함께 제공된다.

2. 실험 분석 방법 및 예시

2.1 각 품질특성치에 대한 기본모형을 위한 설명변수 선택

품질특성치를 y, 제어요인들을 $x_1, \cdots, x_a$(고정된 상수들로 가정함), 잡음요인들을 $z_1, \cdots, z_b$(각각의 평균이 0이고 분산이 σ_z^2이며 서로 상관되어 있지 않은 확률변수들로 가정함)이라 하자. 파라미터 설계가 다구찌식 교차배열이든 반응표면적 결합배열이든 각 품질특성치에 대하여는 다음과 같은 기본모형이 적합될 수 있다.

$$y = f(x_1, \cdots, x_a) + \sum_{i=1}^{b} \delta_i z_i + \sum_{i=1}^{a} \sum_{j=1}^{b} \lambda_{ij} x_i z_j + \varepsilon$$

여기서

$f(x_1, \cdots, x_a) = (x_1, \cdots, x_a)$의 다항식

$\varepsilon =$ 오차항으로 평균 0, 분산 σ_ε^2이며 $z_1, \cdots, z_b$와 상관되어 있지 않음.

기존문헌에서는 $f(x_1, \cdots, x_a)$ 로 2차 다항모형(1차항들, 상호작용항들, 그리고 제곱항들로 이루어짐)이 사용되고 있는데, 다구찌의 교차배열의 경우는 일반적으로 반응표면적 결합배열의 경우에 비하면 상호작용항들의 개수가 더 작다.

우리의 경우에서는 실험설계로 반응표면적 결합배열을 사용한다.

먼저 $f(x_1, \cdots, x_a)$ 로 완전(full) 2차 다항모형을 사용하여 기본모형을 적합시킨 후 적합결여 검정을 실시한다.

적합결여 검정 결과 2차 모형의 적합결여가 유의하지 않으면, 2차 모형 항들을 대상으로 변수선택을 실시하여 선택된 변수들만 들어 있는 축소(reduced) 2차 모형을 찾는다.

적합결여 검정 결과 2차 모형의 적합결여가 유의하면, 3차 모형 항들을 대상으로 변수선택을 실시하여 선택된 변수들만 들어 있는 축소 3차 모형을 찾는다.

모형선택방법으로는 Cp가 가장 작은 모형을 찾는 방법과 수정결정계수(adjusted R^2)가 가장 큰 모형을 찾는 방법을 사용하는데, 최소 Cp 모형과 최대 수정결정계수 모형이 동일하지 않으면 두 모형 중 주어진 상황에서 더 적절하다고 판단되는 모형을 선택하는 방법을 사용한다. 최종적으로 선택된 모형에 대하여도 적합결여 검정을 실시하여 분석에 사용될 모형의 타당성을 확인하는 것이 바람직하다.

이 절차를 각 품질특성치에 적용하여 모형에 들어갈 적절한 항(項)들을 찾아 둔다.

기존 문헌에서는 적합결여가 있을 때에는 고차 다항 모형을 고려하는 것보다는 변수변환이나 다른 형태의 모형을 사용하는 것을 제안(Box, Hunter and Hunter, 1978, p. 523)하고 있지만, 2차 모형의 적합결여를 유의하지 않게 하는 변수변환이 존재하지 않을 수도 있고 사용할 만한 다른 형태의 모형이 어떤 것인지를 알지 못할 수도 있으므로, 2차 모형의 적합결여가 유의할 때 3차 이상의 모형을 고려하는 것은 시도할 만한 일이라고 할 수 있겠다. 이런 경우에 적합결여가 유의하지 않은 3차 모형은 적절한 모형이라고 본다.

[예 2.1] Khuri and Cornell (1987)의 282 페이지에는 설명변수가 다섯 개이고 반응변수가 세 개인 CCD 실험 데이터가 있다. 여기서 앞의 설명변수 세 개를 제어변수들(X1, X2, X3)이라고 하고, 뒤의 설명변수 두 개를 잡음변수들(Z1, Z2)이라고 하자. 그리고 반응변수 세 개 중 뒤의 두 개를 취하여 각각 100으로 나누어 Y1, Y2로 놓자. 이 실험 데이타에서의 실험설계를 <표 2>에서의 설계로 하기 위하여 Khuri and Cornell(1987)의 282 페이지에 있는 데이타의 행들 중 불필요한 행들을 제거하여 <표 3>을 얻었다. 이제 이 데이타를 이용하여 우리의 분석 절차들을 예시하고자 한다.

< 표 3 > Khuri and Cornell (1987, p. 282)에서 발췌한 데이터

처리	X1	X2	X3	Z1	Z2	Y1	Y2
1	-1	-1	-1	-1	1	10.82	0.814
2	1	-1	-1	-1	-1	8.24	0.696
3	-1	1	-1	-1	-1	9.53	1.050
4	1	1	-1	-1	1	7.59	0.812
5	-1	-1	1	-1	-1	11.63	0.808
6	1	-1	1	-1	1	8.39	0.763
7	-1	1	1	-1	1	13.43	1.030
8	1	1	1	-1	-1	7.36	0.769
9	-1	-1	-1	1	-1	10.27	0.872
10	1	-1	-1	1	1	8.36	0.740
11	-1	1	-1	1	1	12.72	0.985
12	1	1	-1	1	-1	8.25	0.941
13	-1	-1	1	1	1	13.63	0.959
14	1	-1	1	1	-1	8.55	0.768
15	-1	1	1	1	-1	12.84	1.000
16	1	1	1	1	1	8.51	1.040
17	-2	0	0	0	0	12.83	1.000
18	2	0	0	0	0	6.51	0.505
19	0	-2	0	0	0	12.17	0.712
20	0	2	0	0	0	9.82	1.010
21	0	0	-2	0	0	8.84	0.858
22	0	0	2	0	0	11.47	1.030
23	0	0	0	0	0	11.79	1.040
23	0	0	0	0	0	11.83	1.070
23	0	0	0	0	0	11.20	1.040
23	0	0	0	0	0	11.80	1.010

먼저 Y1에 관한 2차 모형을 적합시키고 이 모형의 적합결여 검정을 실시하는 SAS 코드가 아래에 나와 있다.

```
OPTIONS CENTER LS=82 PS=55 NODATE PAGENO=1;
DATA A; INPUT TRT X1 X2 X3 Z1 Z2 Y1 Y2; LACKFIT=TRT;
    X1X2=X1*X2; X1X3=X1*X3; X2X3=X2*X3; X1X1=X1*X1; X2X2=X2*X2; X3X3=X3*X3;
    X1X1X1=X1X1*X1; X2X2X2=X2X2*X2; X3X3X3=X3X3*X3;
    X1Z1=X1*Z1; X1Z2=X1*Z2; X2Z1=X2*Z1; X2Z2=X2*Z2; X3Z1=X3*Z1; X3Z2=X3*Z2;
LINES;
 1   -1   -1   -1   -1    1   10.82   0.814
 2    1   -1   -1   -1   -1    8.24   0.696
 ⋮
data lines
 ⋮
23    0    0    0    0    0   11.80   1.010
 ;
TITLE 'Second-Order Model for Y1';
PROC REG DATA=A;
    MODEL Y1=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3
             Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 / VIF;
RUN;
```

```
TITLE 'Test of Lack of Fit of Second-Order Model for Y1';
PROC GLM DATA=A;
     CLASS LACKFIT;
     MODEL Y1=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3
              Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 LACKFIT / SS3;
RUN;
```

Y1에 관한 2차 모형의 적합결여 검정결과를 나타내는 SAS 출력을 아래에 제공한다. 적합결여가 유의함을 알 수 있다.

Test of Lack of Fit of Second-Order Model for Y1

Source	DF	Type III SS	Mean Square	F Value	Pr > F
LACKFIT	5	4.62427727	0.92485545	10.02	0.0433

이제 Y1에 관해서 3차 모형 항들을 대상으로 변수선택을 실시한다. 이를 위한 SAS 코드는 다음과 같다.

```
TITLE 'Selection of Variables of Third-Order Model for Y1';
PROC REG DATA=A;
     MODEL Y1=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 X3X3X3
              Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 / SELECTION=CP ADJRSQ MSE BEST=20;
     MODEL Y1=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 X3X3X3
              Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 / SELECTION=ADJRSQ MSE CP BEST=20;
     MODEL Y1=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 X3X3X3
              Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 / SELECTION=MAXR;
RUN;
```

위 SAS 코드의 출력 중 최소 Cp를 갖는 모형의 항들과 최대 수정결정계수를 갖는 모형의 항들을 제시하는 부분은 다음과 같다.

```
                 Selection of Variables of Third-Order Model for Y1

N = 26      Regression Models for Dependent Variable: Y1

    C(p) R-square          Adj       MSE Variables in Model
                    In     Rsq

 6.6902 0.961090 11 0.930517  0.30445 X1 X3 X1X2 X1X3 X1X1 X2X2 X3X3 X2X2X2 Z1 Z2
                                      X1Z2
```

```
Adj R-square        C(p)     MSE Variables in Model
Rsq            In

0.937363 0.972440 14   10.1552   0.27445 X1 X2 X3 X1X2 X1X3 X1X1 X2X2 X3X3 X1X1X1
                                         X2X2X2 Z1 Z2 X1Z2 X2Z1
```

위 두 모형을 SELECTION=MAXR(maximum R^2 improvement technique)의 출력을 통하여 비교하여 보고, Z1이 관련된 상호작용도 있고 Z2가 관련된 상호작용도 있는 최대 수정결정계수 모형을 선택하기로 한다. (최소 Cp 모형에는 Z1과 제어요인간의 상호작용 항이 없는데, 이 경우에는 제어요인을 이용하여 Z1에 의한 변동을 통제하는 것이 불가능하다.) 선택된 모형의 적합결여 검정을 위한 SAS 코드는 다음과 같다.

```
TITLE 'Test of Lack of Fit of Third-Order Model for Y1';
PROC GLM DATA=A;
    CLASS LACKFIT;
    MODEL Y1=X1 X2 X3 X1X2 X1X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2
           Z1 Z2 X1Z2 X2Z1 LACKFIT/ SS3;
RUN;
```

Y1에 관한 선택된 3차 모형의 적합결여 검정결과를 나타내는 SAS 출력은 아래와 같다. 여기에서 적합결여가 유의하지 않음을 볼 수 있다.

Test of Lack of Fit of Third-Order Model for Y1

Source	DF	Type III SS	Mean Square	F Value	Pr > F
LACKFIT	8	2.74203561	0.34275445	3.71	0.1541

다음으로 Y2에 관한 2차 모형을 적합시키고 이 모형의 적합결여 검정을 실시하는 SAS 코드를 제공한다.

```
TITLE 'Second-Order Model for Y2';
PROC REG DATA=A;
    MODEL Y2=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3
           Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 / VIF;
RUN;

TITLE 'Test of Lack of Fit of Second-Order Model for Y2';
PROC GLM DATA=A;
```

```
    CLASS LACKFIT;
    MODEL Y2=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3
            Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 LACKFIT / SS3;
RUN;
```

Y2에 관한 2차 모형의 적합결여 검정결과를 나타내는 SAS 출력은 아래와 같다. 여기에서 적합결여가 유의함을 알 수 있다.

Test of Lack of Fit of Second-Order Model for Y2

Source	DF	Type III SS	Mean Square	F Value	Pr > F
LACKFIT	5	0.02770413	0.00554083	9.23	0.0484

이제 Y2에 관해서 3차 모형 항들을 대상으로 변수선택을 실시한다. 이를 위한 SAS 코드는 다음과 같다.

```
TITLE 'Selection of Variables of Third-Order Model for Y2';
PROC REG DATA=A;
    MODEL Y2=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 X3X3X3
            Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 / SELECTION=CP ADJRSQ MSE BEST=20;
    MODEL Y2=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 X3X3X3
            Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 / SELECTION=ADJRSQ MSE CP BEST=20;
    MODEL Y2=X1 X2 X3 X1X2 X1X3 X2X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 X3X3X3
            Z1 Z2 X1Z1 X1Z2 X2Z1 X2Z2 X3Z1 X3Z2 / SELECTION=MAXR;
RUN;
```

위 SAS 코드의 출력 중 최소 Cp를 갖는 모형의 항들과 최대 수정결정계수를 갖는 모형의 항들을 제시하는 부분을 다음에 제공한다.

```
              Selection of Variables of Third-Order Model for Y2

N = 26     Regression Models for Dependent Variable: Y2

   C(p) R-square         Adj       MSE Variables in Model
                  In     Rsq

 7.8639 0.986529 12 0.974094 0.000558 X1 X2 X1X1 X2X2 X3X3 X1X1X1 X3X3X3 Z1 Z2
                                      X1Z1 X3Z1 X3Z2
-------------------------------------------------------------------------------
   Adj R-square          C(p)      MSE Variables in Model
   Rsq            In
```

0.977497	0.990099	14	9.7800	0.000485	X1 X2 X2X3 X1X1 X2X2 X3X3 X1X1X1 X3X3X3 Z1 Z2 X1Z1 X1Z2 X3Z1 X3Z2

위 두 모형을 SELECTION=MAXR의 출력을 통하여 비교하여 보고, 제어요인과 잡음요인 간의 상호작용 항들이 더 많은 최대 수정결정결수 모형을 선택하기로 한다. 선택된 모형의 적합결여 검정을 위한 SAS 코드는 다음과 같다.

```
TITLE 'Test of Lack of Fit of Third-Order Model for Y2';
PROC GLM DATA=A;
    CLASS LACKFIT;
    MODEL Y2=X1 X2 X2X3 X1X1 X2X2 X3X3 X1X1X1 X3X3X3
             Z1 Z2 X1Z1 X1Z2 X3Z1 X3Z2 LACKFIT/ SS3;
RUN;
```

Y2에 관한 선택된 3차 모형의 적합결여 검정결과를 나타내는 SAS 출력을 아래에 제공한다. 적합결여가 유의하지 않음을 알 수 있다.

Test of Lack of Fit of Third-Order Model for Y2

Source	DF	Type III SS	Mean Square	F Value	Pr > F
LACKFIT	8	0.00353140	0.00044142	0.74	0.6772

2.2 전체 관측치에 대한 기본모형으로서의 SUR 모형의 FGLS 추정

각 품질특성치에 대한 기본모형에 들어갈 항들이 모두 같지는 않은 것이 보통이다. 그리고, 동일한 실험조건 하에서 측정되는 품질특성치들간에는 상관(correlation)이 존재할 수 있다. 품질특성치의 개수가 k이고 실험시행 수가 n일 때, $\boldsymbol{y}_i\,(n\times 1)$를 i번째 품질특성의 관측치 n개로 이루어지는 열 벡터라고 하고, $\boldsymbol{W}_i$를 $\boldsymbol{y}_i$를 설명하는 변수들의 열 벡터들로 이루어진 행렬이라고 하며, $\boldsymbol{0}$을 0들로 이루어진 행렬이라고 하면, 모든 품질특성들의 관측치들의 열 벡터들을 세로로 연결한 전체 관측벡터 $\boldsymbol{y}\,(kn\times 1)$는 다음과 같은 SUR 모형으로 나타낼 수 있다.

$$\boldsymbol{y} = \boldsymbol{W}\boldsymbol{\alpha} + \boldsymbol{\varepsilon}$$

즉,

$$\begin{bmatrix} \boldsymbol{y}_1 \\ \boldsymbol{y}_2 \\ \vdots \\ \boldsymbol{y}_k \end{bmatrix} = \begin{bmatrix} \boldsymbol{W}_1 & \boldsymbol{0} & \cdots & \boldsymbol{0} \\ \boldsymbol{0} & \boldsymbol{W}_2 & \cdots & \boldsymbol{0} \\ \vdots & \vdots & \ddots & \vdots \\ \boldsymbol{0} & \boldsymbol{0} & \cdots & \boldsymbol{W}_k \end{bmatrix} \begin{bmatrix} \boldsymbol{\alpha}_1 \\ \boldsymbol{\alpha}_2 \\ \vdots \\ \boldsymbol{\alpha}_k \end{bmatrix} + \begin{bmatrix} \boldsymbol{\varepsilon}_1 \\ \boldsymbol{\varepsilon}_2 \\ \vdots \\ \boldsymbol{\varepsilon}_k \end{bmatrix}$$

기본모형에서는 일단 $\boldsymbol{W}$를 주어진 것으로 보기 때문에 $\mathrm{Cov}(\boldsymbol{y}) = \mathrm{Cov}(\boldsymbol{\varepsilon})$로 된다. 동일한 실험조건 하에서 i번째 품질특성치의 분산을 $\mathrm{Var}(y_i) = \sigma_i^2$이라 하고, i번째 품질특성치와 j번째 품질특성치 간의 공분산을 $\mathrm{Cov}(y_i, y_j) = \sigma_{ij}$라고 하면, 전체오차벡터 $\boldsymbol{\varepsilon}$의 공분산행렬은 다음과 같이 된다.

$$\mathrm{Cov}(\boldsymbol{\varepsilon}) = \mathrm{Cov}\begin{bmatrix} \boldsymbol{\varepsilon}_1 \\ \boldsymbol{\varepsilon}_2 \\ \vdots \\ \boldsymbol{\varepsilon}_k \end{bmatrix} = \begin{bmatrix} \sigma_1^2\,\mathbf{I}_n & \sigma_{12}\,\mathbf{I}_n & \cdots & \sigma_{1k}\,\mathbf{I}_n \\ \sigma_{12}\,\mathbf{I}_n & \sigma_2^2\,\mathbf{I}_n & \cdots & \sigma_{2k}\,\mathbf{I}_n \\ \vdots & \vdots & \ddots & \vdots \\ \sigma_{1k}\,\mathbf{I}_n & \sigma_{2k}\,\mathbf{I}_n & \cdots & \sigma_k^2\,\mathbf{I}_n \end{bmatrix}$$

$$= \begin{bmatrix} \sigma_{11} & \sigma_{12} & \cdots & \sigma_{1k} \\ \sigma_{12} & \sigma_{22} & \cdots & \sigma_{2k} \\ \vdots & \vdots & \ddots & \vdots \\ \sigma_{1k} & \sigma_{2k} & \cdots & \sigma_{kk} \end{bmatrix} \otimes \mathbf{I}_n = \boldsymbol{\Sigma} \otimes \mathbf{I}_n \tag{1}$$

여기서 $\otimes$는 Kronecker product를 나타낸다. 이 SUR 모형에서의 전체회귀계수 벡터 $\boldsymbol{\alpha}$의 FGLS (feasible generalized least squares) 추정치는

$$\hat{\boldsymbol{\alpha}} = [\,\boldsymbol{W}'(\,\hat{\boldsymbol{\Sigma}}^{-1} \otimes \mathbf{I}_n)\,\boldsymbol{W}]^{-1}\,\boldsymbol{W}'(\,\hat{\boldsymbol{\Sigma}}^{-1} \otimes \mathbf{I}_n)\,\boldsymbol{y}$$

인데, SAS의 PROC SYSLIN을 사용하여 구할 수 있다(박유성, 송석헌, 1998).

[예 2.2] <표 3>의 데이터에 예 2.1에서 구한 각 품질특성치에 관한 모형의 항들에 대하여 OLS(ordinary least squares) 추정치들과 SUR 모형에서의 FGLS 추정치들을 나란히 인쇄시키는 SAS 코드를 아래에 싣는다.

```
TITLE 'Seemingly Unrelated Regressions (SUR) for Y1 and Y2';
PROC SYSLIN DATA=A OUTEST=EST SUR;
    Y1: MODEL Y1=X1 X2 X3 X1X2 X1X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 Z1 Z2 X1Z2 X2Z1;
    Y2: MODEL Y2=X1 X2 X2X3 X1X1 X2X2 X3X3 X1X1X1 X3X3X3 Z1 Z2 X1Z1 X1Z2 X3Z1 X3Z2;
RUN;

TITLE 'Comparison of OLS and FGLS Estimates';
```

```
DATA ESTPART; SET EST; DROP _TYPE_ _MODEL_ _DEPVAR_ Y1 Y2;
PROC TRANSPOSE DATA=ESTPART OUT=TRANSEST; VAR _ALL_;
RUN;

DATA TRANSEST; SET TRANSEST;
  LABEL _NAME_='Model Term'
      COL1='OLS Estimate in Model for Y1'       COL2='OLS Estimate in Model for Y2'
      COL3='FGLS Estimate in SUR Model for Y1' COL4='FGLS Estimate in SUR Model for Y2';

PROC PRINT DATA=TRANSEST NOOBS LABEL;
    VAR _NAME_ COL1 COL3 COL2 COL4; FORMAT COL1 COL3 COL2 COL4 10.6;
RUN;
```

위 SAS 프로그램의 출력 중 앞의 (1)에서의 Σ의 추정치와 상관행렬의 추정치를 아래에 싣는다.

```
Seemingly Unrelated Regressions (SUR) for Y1 and Y2

                SYSLIN Procedure
      Seemingly Unrelated Regression Estimation

              Cross Model Covariance

      Sigma              Y1                Y2

      Y1       0.2744486915      0.0064477365
      Y2       0.0064477365      0.0004846724
              Cross Model Correlation

      Corr               Y1                Y2

      Y1                  1      0.5590524413
      Y2       0.5590524413                 1
```

위 출력으로부터 Corr(Y1, Y2)=0.559임을 알 수 있다. Y1과 Y2에 관한 모형들의 계수들의 OLS 추정치들과 FGLS 추정치들은 다음과 같다. (OLS 추정치들은 Corr(Y1, Y2)를 무시한 추정치들이고 FGLS 추정치들은 Corr(Y1, Y2)를 고려한 추정치들이다.)

```
                Comparison of OLS and FGLS Estimates

                OLS          FGLS          OLS          FGLS
              Estimate    Estimate in    Estimate    Estimate in
   Model      in Model    SUR Model      in Model    SUR Model
   Term        for Y1       for Y1        for Y2       for Y2
```

SIGMA	0.523879	0.534403	0.022015	0.022454
INTERCEP	11.432727	11.432727	1.032205	1.032205
X1	-1.941667	-1.941667	-0.041167	-0.041167
X2	0.224167	0.215852	0.075125	0.075125
X3	0.575833	0.539412	.	.
X1X2	-0.250000	-0.235865	.	.
X1X3	-0.488750	-0.551109	.	.
X1X1	-0.551818	-0.551818	-0.073824	-0.073824
X2X2	-0.220568	-0.220568	-0.046699	-0.046699
X3X3	-0.430568	-0.430568	-0.025949	-0.025949
X1X1X1	0.090417	0.090417	-0.020646	-0.020646
X2X2X2	-0.202917	-0.198759	.	.
Z1	0.383750	0.383750	0.035188	0.035188
Z2	0.423750	0.423750	0.014938	0.014938
X1Z2	-0.367500	-0.367500	0.007688	0.007688
X2Z1	0.167500	0.128422	.	.
X2X3	.	.	-0.007813	-0.007137
X3X3X3	.	.	0.011132	0.010527
X1Z1	.	.	0.020938	0.023815
X3Z1	.	.	0.014438	0.015465
X3Z2	.	.	0.040938	0.040380

위의 출력에서 _SIGMA_는 각 회귀모형의 오차항의 표준편차를 나타낸다. 우리는 SUR 모형에서의 FGLS 추정치들을 후속 분석에 사용한다.

2.3 각 품질특성치의 평균과 분산의 동시 모형화

각 품질특성치 y에 대하여,

$$\boldsymbol{x} = (x_1\ ,\ x_2\ ,\ \cdots,\ x_a)'\ ,\quad \boldsymbol{z} = (z_1\ ,\ z_2\ ,\ \cdots,\ z_b)'\ ,$$

$$\boldsymbol{\delta} = (\delta_1\ ,\ \delta_2\ ,\ \cdots,\ \delta_b)'\ ,\quad \boldsymbol{\Lambda} = \begin{bmatrix} \lambda_{11} & \lambda_{12} & \cdots & \lambda_{1b} \\ \lambda_{21} & \lambda_{22} & \cdots & \lambda_{2b} \\ \vdots & \vdots & \ddots & \vdots \\ \lambda_{a1} & \lambda_{a2} & \cdots & \lambda_{ab} \end{bmatrix},$$

$$\mathrm{E}(\boldsymbol{z}) = \begin{bmatrix} 0 \\ 0 \\ \vdots \\ 0 \end{bmatrix},\quad \mathbf{V} = \mathrm{Cov}(\boldsymbol{z}) = \begin{bmatrix} \sigma_z^2 & 0 & \cdots & 0 \\ 0 & \sigma_z^2 & \cdots & 0 \\ \vdots & \vdots & \ddots & \\ 0 & 0 & \cdots & \sigma_z^2 \end{bmatrix} = \sigma_z^2\ \mathbf{I}_b\ ,$$

$$\mathrm{E}(\varepsilon) = 0\ ,\quad \mathrm{Var}(\varepsilon) = \sigma_\varepsilon^2$$

이라 할 때, y에 대한 기본 모형은 다음과 같이 표시된다.

$$\begin{aligned} y &= (x_1, \cdots, x_a)\text{의 다항식} + \boldsymbol{\delta}'\boldsymbol{z} + \boldsymbol{x}'\boldsymbol{\Lambda}\boldsymbol{z} + \varepsilon \\ &= (x_1, \cdots, x_a)\text{의 다항식} + (\boldsymbol{\delta}' + \boldsymbol{x}'\boldsymbol{\Lambda})\boldsymbol{z} + \varepsilon \end{aligned} \tag{2}$$

모형 (2)으로부터 다음과 같이 y의 평균 모형과 분산 모형이 구해진다.

$$\mathrm{E}(y) = \mu_y = (x_1, \cdots, x_a)\text{의 다항식} \tag{3}$$

$$\begin{aligned} \mathrm{Var}(y) = \sigma_y^2 &= \sigma_\varepsilon^2 + (\boldsymbol{\delta}' + \boldsymbol{x}'\boldsymbol{\Lambda})\,\mathbf{V}\,(\boldsymbol{\delta}' + \boldsymbol{x}'\boldsymbol{\Lambda})' \\ &= \sigma_\varepsilon^2 + \boldsymbol{\delta}'\,\mathbf{V}\,\boldsymbol{\delta} + 2\,\boldsymbol{\delta}'\,\mathbf{V}\,\boldsymbol{\Lambda}'\boldsymbol{x} + \boldsymbol{x}'\boldsymbol{\Lambda}\,\mathbf{V}\,\boldsymbol{\Lambda}'\boldsymbol{x} \\ &= \sigma_\varepsilon^2 + \sigma_z^2(\boldsymbol{\delta}'\boldsymbol{\delta} + 2\,\boldsymbol{\delta}'\boldsymbol{\Lambda}'\boldsymbol{x} + \boldsymbol{x}'\boldsymbol{\Lambda}\boldsymbol{\Lambda}'\boldsymbol{x}) \\ &= \sigma_\varepsilon^2 + \sigma_z^2(\beta_0 + \sum_{i=1}^{a}\beta_i x_i + \sum_{i=1}^{a-1}\sum_{j=i+1}^{a}\beta_{ij}x_i x_j + \sum_{i=1}^{a}\beta_{ii}x_i^2) \end{aligned} \tag{4}$$

분산 대신에 표준편차가 자주 사용되므로 표준편차 모형을 위의 분산 모형의 제곱근으로 다음과 같이 정의한다.

$$\mathrm{SD}(y) = \sigma_y = [\,\sigma_\varepsilon^2 + \sigma_z^2(\beta_0 + \sum_{i=1}^{a}\beta_i x_i + \sum_{i=1}^{a-1}\sum_{j=i+1}^{a}\beta_{ij}x_i x_j + \sum_{i=1}^{a}\beta_{ii}x_i^2)]^{1/2} \tag{5}$$

z_i는 낮은 값이 -1, 중간 값이 0, 높은 값이 1이 되도록 코드화된 i번째 잡음요인인데, (4)과 (5)에서 σ_z^2(noise variance)의 값은, 각 z_i의 수준 0이 각 잡음요인의 평균에 대응되고 각 z_i의 수준 ± 1이 [(각 잡음요인의 평균) $\pm$ c(각 잡음요인의 표준편차)]에 대응될 때, c^{-2}으로 놓는다(Myers and Montgomery, 1995, p. 493). 각 잡음요인의 평균과 표준편차는 사전정보나 과거의 경험에 의해 주어지는 것으로 한다. (3)과 (4)의 모형은 각 모형 항의 계수들을 기본모형 (2)의 모형 항들의 FGLS 추정치들로 대치함으로 추정된다. (4)에서의 σ_ε^2(error variance)의 추정치는 기본모형 (2)의 오차항의 분산의 추정치이다.

[예 2.3] 예 2.2에서 FGLS 방법으로 추정된 Y1과 Y2의 기본모형들에 의해 Y1, Y2 각각의 평균 모형과 분산 모형을 추정하는 SAS 코드는 다음과 같다. 이 SAS 코드는 예 2.2의 SAS 코드에 이어지는 것이다.

```
DATA COEFF1; SET EST; IF _TYPE_='SUR' AND _DEPVAR_='Y1';
   KEEP INTERCEP X1 X2 X3 X1X2 X1X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 Z1 Z2 X1Z2 X2Z1;

DATA MSE1; SET EST; IF _TYPE_='SUR' AND _DEPVAR_='Y1';
   MSE1=_SIGMA_*_SIGMA_; KEEP MSE1;

DATA COEFF2; SET EST; IF _TYPE_='SUR' AND _DEPVAR_='Y2';
   KEEP INTERCEP X1 X2 X2X3 X1X1 X2X2 X3X3 X1X1X1 X3X3X3 Z1 Z2 X1Z1 X1Z2 X3Z1 X3Z2;

DATA MSE2; SET EST; IF _TYPE_='SUR' AND _DEPVAR_='Y2';
   MSE2=_SIGMA_*_SIGMA_; KEEP MSE2;

TITLE;

PROC IML;

  USE COEFF1
      VAR {INTERCEP X1 X2 X3 X1X2 X1X3 X1X1 X2X2 X3X3 X1X1X1 X2X2X2 Z1 Z2 X1Z2
    X2Z1}; READ ALL INTO COEFF1;
  USE COEFF2
      VAR {INTERCEP X1 X2 X2X3 X1X1 X2X2 X3X3 X1X1X1 X3X3X3 Z1 Z2 X1Z1 X1Z2 X3Z1
    X3Z2}; READ ALL INTO COEFF2;

  USE MSE1; READ ALL INTO MSE1;
  USE MSE2; READ ALL INTO MSE2;

  Y1INT     = COEFF1[ 1];
  Y1X1      = COEFF1[ 2];  Y1X2      = COEFF1[ 3];  Y1X3      = COEFF1[ 4];
  Y1X1X2    = COEFF1[ 5];  Y1X1X3    = COEFF1[ 6];  Y1X2X3    = 0;
  Y1X1X1    = COEFF1[ 7];  Y1X2X2    = COEFF1[ 8];  Y1X3X3    = COEFF1[ 9];
  Y1X1X1X1 = COEFF1[10];  Y1X2X2X2 = COEFF1[11];  Y1X3X3X3 = 0;
  Y1Z1      = COEFF1[12];  Y1Z2      = COEFF1[13];
  Y1X1Z1    = 0;           Y1X1Z2    = COEFF1[14];
  Y1X2Z1    = COEFF1[15];  Y1X2Z2    = 0;
  Y1X3Z1    = 0;           Y1X3Z2    = 0;

  Y2INT     = COEFF2[ 1];
  Y2X1      = COEFF2[ 2];  Y2X2      = COEFF2[ 3];  Y2X3      = 0;
  Y2X1X2    = 0;           Y2X1X3    = 0;           Y2X2X3    = COEFF2[ 4];
  Y2X1X1    = COEFF2[ 5];  Y2X2X2    = COEFF2[ 6];  Y2X3X3    = COEFF2[ 7];
  Y2X1X1X1 = COEFF2[ 8];  Y2X2X2X2 = 0;           Y2X3X3X3 = COEFF2[ 9];
  Y2Z1      = COEFF2[10];  Y2Z2      = COEFF2[11];
  Y2X1Z1    = COEFF2[12];  Y2X1Z2    = COEFF2[13];
  Y2X2Z1    = 0;           Y2X2Z2    = 0;
  Y2X3Z1    = COEFF2[14];  Y2X3Z2    = COEFF2[15];

  DELTA1 = Y1Z1//Y1Z2;
  DELTA2 = Y2Z1//Y2Z2;
  LAMBDA1=(Y1X1Z1||Y1X1Z2)//(Y1X2Z1||Y1X2Z2)//(Y1X3Z1||Y1X3Z2);
  LAMBDA2=(Y2X1Z1||Y2X1Z2)//(Y2X2Z1||Y2X2Z2)//(Y2X3Z1||Y2X3Z2);

  M1 = Y1INT // Y1X1 // Y1X2 // Y1X3 // Y1X1X2 // Y1X1X3 // Y1X2X3 //
      Y1X1X1 // Y1X2X2 // Y1X3X3 // Y1X1X1X1 // Y1X2X2X2 // Y1X3X3X3;
```

```
NAMEM1 = {M1INT, M1X1, M1X2, M1X3, M1X1X2, M1X1X3, M1X2X3,
          M1X1X1, M1X2X2, M1X3X3, M1X1X1X1, M1X2X2X2, M1X3X3X3};

TERM = {'intercept', 'X1', 'X2', 'X3', 'X1*X2', 'X1*X3', 'X2*X3',
        'X1*X1', 'X2*X2', 'X3*X3', 'X1*X1*X1', 'X2*X2*X2', 'X3*X3*X3'};

M2 = Y2INT // Y2X1 // Y2X2 // Y2X3 // Y2X1X2 // Y2X1X3 // Y2X2X3 //
     Y2X1X1 // Y2X2X2 // Y2X3X3 // Y2X1X1X1 // Y2X2X2X2 // Y2X3X3X3;

NAMEM2 = {M2INT, M2X1, M2X2, M2X3, M2X1X2, M2X1X3, M2X2X3,
          M2X1X1, M2X2X2, M2X3X3, M2X1X1X1, M2X2X2X2, M2X3X3X3};

MEAN = M1||M2;

PRINT 'Parameter Estimates in Mean Models for Y1 and Y2',
      MEAN[ROWNAME=TERM COLNAME={Y1 Y2} FORMAT=10.6],;

R=2; SIGMAZ=1; VZ=SIGMAZ*SIGMAZ;
COVZ=VZ#I(R);

V1INT = DELTA1`*COVZ*DELTA1;
V1L   = 2#(DELTA1`*COVZ*LAMBDA1`)`;
V1QF  = LAMBDA1*COVZ*LAMBDA1`;

V2INT = DELTA2`*COVZ*DELTA2;
V2L   = 2#(DELTA2`*COVZ*LAMBDA2`)`;
V2QF  = LAMBDA2*COVZ*LAMBDA2`;

V1X1  = V1L[1];        V1X2  = V1L[2];        V1X3  = V1L[3];
V1X1X2=2#V1QF[1,2];  V1X1X3=2#V1QF[1,3];  V1X2X3=2#V1QF[2,3];
V1X1X1=V1QF[1,1];    V1X2X2=V1QF[2,2];    V1X3X3=V1QF[3,3];

V2X1  =V2L[1];         V2X2  =V2L[2];         V2X3  =V2L[3];
V2X1X2=2#V2QF[1,2];  V2X1X3=2#V2QF[1,3];  V2X2X3=2#V2QF[2,3];
V2X1X1=V2QF[1,1];    V2X2X2=V2QF[2,2];    V2X3X3=V2QF[3,3];

V1 = MSE1 // VZ // V1INT // V1X1 // V1X2 // V1X3 //
     V1X1X2 // V1X1X3 // V1X2X3 // V1X1X1 // V1X2X2 // V1X3X3;

NAMEV1 = {MSE1, VZ, V1INT, V1X1, V1X2, V1X3, V1X1X2, V1X1X3, V1X2X3,
          V1X1X1, V1X2X2, V1X3X3};

TERM = {'error variance', 'noise variance', 'intercept', 'X1', 'X2', 'X3',
        'X1*X2', 'X1*X3', 'X2*X3', 'X1*X1', 'X2*X2', 'X3*X3'};

V2 = MSE2 // VZ // V2INT // V2X1 // V2X2 // V2X3 // V2X1X2 // V2X1X3 // V2X2X3//
     V2X1X1 // V2X2X2 // V2X3X3;

NAMEV2 = {MSE2, VZ, V2INT, V2X1, V2X2, V2X3, V2X1X2, V2X1X3, V2X2X3,
          V2X1X1, V2X2X2, V2X3X3};

VARIANCE = V1||V2;

PRINT 'Parameter Estimates in Variance Models for Y1 and Y2',
      VARIANCE[ROWNAME=TERM COLNAME={Y1 Y2} FORMAT=10.6],;
```

```
  M1P=M1`; M2P=M2`; NAMEM1P=NAMEM1`; NAMEM2P=NAMEM2`;
  V1P=V1`; V2P=V2`; NAMEV1P=NAMEV1`; NAMEV2P=NAMEV2`;

  CREATE M1P FROM M1P[COLNAME=NAMEM1P]; APPEND FROM M1P;
  CREATE M2P FROM M2P[COLNAME=NAMEM2P]; APPEND FROM M2P;
  CREATE V1P FROM V1P[COLNAME=NAMEV1P]; APPEND FROM V1P;
  CREATE V2P FROM V2P[COLNAME=NAMEV2P]; APPEND FROM V2P;

QUIT;
```

위 SAS 코드에 의한 출력은 다음과 같다. 우리의 예에서는 잡음분산을 $\sigma_z^2 = 1$으로 놓았다. 표준편차 모형은 아래의 분산 모형의 제곱근을 취함으로써 얻어진다.

Parameter Estimates in Mean Models for Y1 and Y2

MEAN	Y1	Y2
intercept	11.432727	1.032205
X1	-1.941667	-0.041167
X2	0.215852	0.075125
X3	0.539412	0.000000
X1*X2	-0.235865	0.000000
X1*X3	-0.551109	0.000000
X2*X3	0.000000	-0.007137
X1*X1	-0.551818	-0.073824
X2*X2	-0.220568	-0.046699
X3*X3	-0.430568	-0.025949
X1*X1*X1	0.090417	-0.020646
X2*X2*X2	-0.198759	0.000000
X3*X3*X3	0.000000	0.010527

Parameter Estimates in Variance Models for Y1 and Y2

VARIANCE	Y1	Y2
error variance	0.285586	0.000504
noise variance	1.000000	1.000000
intercept	0.326828	0.001461
X1	-0.311456	0.001906
X2	0.098564	0.000000
X3	0.000000	0.002295
X1*X2	0.000000	0.000000
X1*X3	0.000000	0.001357
X2*X3	0.000000	0.000000
X1*X1	0.135056	0.000626
X2*X2	0.016492	0.000000
X3*X3	0.000000	0.001870

2.4 품질특성치들의 평균들과 표준편차들의 동시 최적화

다반응 최적화를 위하여 바람직성 함수(Derringer and Suich, 1980)를 이용하고자 한다. 먼저 망대(望大) 특성, 망소(望小) 특성, 그리고 망목(望目) 특성의 각 경우에 있어서의 바람직성 함수를 소개한다. 바람직성 함수는 반응의 추정치가 만족스러운 값일 때는 1의 값을 갖고 반응의 추정치가 받아들일 수 없는 값일 때는 0의 값을 갖도록 되어 있다. 다음에서 반응의 추정치를 $\hat{y}$로 표기한다.

(1) 망대(larger the better) 특성을 위한 바람직성 함수

$\hat{y}$이 low 이하이면 받아들일 수 없고 high 이상이면 만족할 경우의 바람직성 함수는 다음과 같다.

$$d(\hat{y}) = d_L(\hat{y}) = \begin{cases} 0, & \hat{y} \le \text{low} \\ [(\hat{y} - \text{low})/(\text{high} - \text{low})]^{\text{weight}}, & \text{low} < \hat{y} < \text{high} \\ 1, & \text{high} \le \hat{y} \end{cases} \tag{6}$$

(2) 망소(smaller the better) 특성을 위한 바람직성 함수

$\hat{y}$이 low 이하이면 만족하고 high 이상이면 받아들일 수 없을 경우의 바람직성 함수는 다음과 같다.

$$d(\hat{y}) = d_S(\hat{y}) = \begin{cases} 1, & \hat{y} \le \text{low} \\ [(\text{high} - \hat{y})/(\text{high} - \text{low})]^{\text{weight}}, & \text{low} < \hat{y} < \text{high} \\ 0, & \text{high} \le \hat{y} \end{cases} \tag{7}$$

(3) 망목(target the best) 특성을 위한 바람직성 함수

$\hat{y}$이 target 이면 만족하고 low 이하이거나 high 이상이면 받아들일 수 없을 경우의 바람직성 함수는 다음과 같다.

$$d(\hat{y}) = d_T(\hat{y}) = \begin{cases} 0, & \hat{y} \le \text{low} \\ [(\hat{y} - \text{low})/(\text{target} - \text{low})]^{\text{weight 1}}, & \text{low} < \hat{y} \le \text{target} \\ [(\text{high} - \hat{y})/(\text{high} - \text{target})]^{\text{weight 2}}, & \text{target} < \hat{y} < \text{high} \\ 0, & \text{high} \le \hat{y} \end{cases} \tag{8}$$

$\hat{y}$이 목표치에 가까워야 하는 것이 더 강하게 요구되면, 연구자는 위의 (6), (7), (8)에서의 멱(power)인 weight, weight 1, 그리고/또는 weight 2를 더 높게 정하게 된다.

2.4.1 평균들의 바람직성 함수

평균 $\widehat{\mu_1}$, $\widehat{\mu_2}$, …, $\widehat{\mu_k}$의 각각에 대하여 적절하게 바람직성 함수

$$d(\widehat{\mu_1}),\ d(\widehat{\mu_2}),\ \cdots,\ d(\widehat{\mu_k})$$

을 정의하고, 이들의 기하평균을 평균들의 바람직성 함수 D_M으로 정의한다.

$$D_M = [\, d(\widehat{\mu_1}) \cdot d(\widehat{\mu_2}) \cdot \cdots \cdot d(\widehat{\mu_k})\,]^{1/k}$$

2.4.2 표준편차들의 바람직성 함수

표준편차 $\widehat{\sigma_1}$, $\widehat{\sigma_2}$, …, $\widehat{\sigma_k}$의 각각에 대하여 적절하게 바람직성 함수

$$d(\widehat{\sigma_1}),\ d(\widehat{\sigma_2}),\ \cdots,\ d(\widehat{\sigma_k})$$

을 정의하고, 이들의 기하평균을 표준편차들의 바람직성 함수 D_S로 정의한다.

$$D_S = [\, d(\widehat{\sigma_1}) \cdot d(\widehat{\sigma_2}) \cdot \cdots \cdot d(\widehat{\sigma_k})\,]^{1/k}$$

2.4.3 평균들과 표준편차들의 종합 바람직성 함수

우리의 목적은 평균들과 표준편차들을 동시에 최적화하는 것이지만, 평균들의 최적화가 중요한 정도와 표준편차들의 최적화가 중요한 정도가 서로 다를 수 있다. 그래서, 다음과 같이 가중치(weight)를 두어 위의 D_M과 D_S를 결합하여 종합 바람직성(overall desirability) 함수 D를 정의한다.

$$D = D_M^{w} \cdot D_S^{(1-w)}$$

여기서 w는 평균들의 최적화에 주어지는 가중치로서 $0 \le w \le 1$의 범위를 갖는다. 우리는 이제 이 D를 최대화하는 제어요인(x_1, x_2, …, x_a)들의 수준값들을 찾는다.

D를 최대화하는 제어요인들의 수준값들을 찾는 방법으로서, 실험구역 내의 격자(grid)점들에서 평균들과 표준편차들의 추정치들을 구하고, 이 추정치들 각각에 대하

여 종합 바람직성 함수 D의 값을 계산한 다음, 이 D의 값들을 크기 순으로 정렬하여 D의 값을 최대화하는 x_1, x_2, …, x_a 각각의 수준값을 찾는 방법을 사용할 수 있다 (이우선, 김영주, 1997).

평균들의 최적화의 중요성이 표준편차들의 중요성보다 더 크면 w를 0.5보다 크게 놓고, 표준편차들의 최적화의 중요성이 표준편차들의 중요성보다 더 크면 w를 0.5보다 작게 놓는다. 극단적으로 $w = 1$ 이면 표준편차들에 관계없이 평균들만을 최적화하게 되고, $w = 0$ 이면 평균들에 관계없이 표준편차들만을 최적화하게 된다.

[예 2.4] 예 2.3에서 얻어진 평균 모형들과 분산 모형들을 이용하여 평균들과 분산들을 최적화하여 보자.

Y1의 평균은 클수록 좋은데, 최소한 8은 되어야 하고 12이상이면 만족스럽다고 하자. 그리고, Y2의 평균은 7.5이면 가장 좋은데, 7보다 작거나 8보다 크면 곤란하다고 하자. 그러면, Y1의 평균을 위한 바람직성 함수로는 d_L 이 사용되고 d_L 에서 $low = 8$, $high = 12$ 가 된다. Y2의 평균을 위한 바람직성 함수로는 d_T 가 사용되고 d_T 에서 $low = 7$, $target = 7.5$, $high = 8$ 이 된다.

Y1의 표준편차는 작을수록 좋은데, Y1의 잡음요인이 아닌 오차항의 표준편차의 추정치가 0.5344(예 2.2의 SAS 출력에 _SIGMA_의 추정치로 나타나 있음)이므로 Y1의 표준편차가 0.6 이하이면 만족스럽고 0.9 이상이면 곤란하다고 하자. 그리고, Y2의 표준편차도 작을수록 좋은데, Y2의 잡음요인이 아닌 오차항의 표준편차가 0.0225(역시 예 2.2의 SAS 출력에 _SIGMA_의 추정치로 나타나 있음)이므로 Y2의 표준편차가 0.025 이하이면 만족스럽고 0.0375 이상이면 곤란하다고 하자. 그러면, Y2의 표준편차를 위한 바람직성 함수로는 d_S가 사용되고 d_S 에서 $low = 0.6$, $high = 0.9$ 가 된다. Y2의 표준편차를 위한 바람직성 함수로도 d_S 가 사용되고 d_S 에서 $low = 0.025$, $high = 0.0375$ 가 된다.

이제 평균들과 표준편차들의 동시 최적화를 위한 SAS 코드를 다음에 제공한다.

```
DATA ALL; MERGE M1P M2P V1P V2P MSE1 MSE2;
    DO X1=-2 TO 2 BY 0.1; DO X2=-2 TO 2 BY 0.1; DO X3=-2 TO 2 BY 0.1;
      MEAN1 = M1INT
          + M1X1      * X1        + M1X2      * X2        + M1X3      * X3
          + M1X1X2    * X1*X2     + M1X1X3    * X1*X3     + M1X2X3    * X2*X3
          + M1X1X1    * X1*X1     + M1X2X2    * X2*X2     + M1X3X3    * X3*X3
          + M1X1X1X1 * X1*X1*X1 + M1X2X2X2 * X2*X2*X2 + M1X3X3X3 * X3*X3*X3;
      MEAN2 = M2INT
          + M2X1      * X1        + M2X2      * X2        + M2X3      * X3
          + M2X1X2    * X1*X2     + M2X1X3    * X1*X3     + M2X2X3    * X2*X3
          + M2X1X1    * X1*X1     + M2X2X2    * X2*X2     + M2X3X3    * X3*X3
          + M2X1X1X1 * X1*X1*X1 + M2X2X2X2 * X2*X2*X2 + M2X3X3X3 * X3*X3*X3;
```

```
        V1 = MSE1 + VZ * (V1INT
                   + V1X1   * X1    + V1X2   * X2    + V1X3   * X3
                   + V1X1X2 * X1*X2 + V1X1X3 * X1*X3 + V1X2X3 * X2*X3
                   + V1X1X1 * X1*X1 + V1X2X2 * X2*X2 + V1X3X3 * X3*X3);
      SD1=SQRT(V1);
      V2 = MSE2 + VZ * (V2INT
                   + V2X1   * X1    + V2X2   * X2    + V2X3   * X3
                   + V2X1X2 * X1*X2 + V2X1X3 * X1*X3 + V2X2X3 * X2*X3
                   + V2X1X1 * X1*X1 + V2X2X2 * X2*X2 + V2X3X3 * X3*X3);
      SD2=SQRT(V2);
      OUTPUT;
    END; END; END;

DATA ALL; SET ALL; KEEP X1 X2 X3 MEAN1 MEAN2 SD1 SD2;

%MACRO MULTIOPT (W_MEAN, W_MEAN1, W1_MEAN2, W2_MEAN2, W_SD1, W_SD2);

DATA D; SET ALL;
        M1LOW=8;  M1HIGH=12;            W_MEAN1 =&W_MEAN1;
        M2LOW=0.7;     M2TARGET=0.75;     M2HIGH=0.8;     W1_MEAN2=&W1_MEAN2;
W2_MEAN2=&W2_MEAN2;

    IF MEAN1  <= M1LOW           THEN DM1=0;
ELSE    IF    M1LOW     <      MEAN1     <     M1HIGH     THEN
DM1=((MEAN1-M1LOW)/(M1HIGH-M1LOW))**W_MEAN1;
ELSE IF M1HIGH <= MEAN1          THEN DM1=1;

    IF MEAN2 <= M2LOW            THEN DM2=0;
ELSE    IF    M2LOW    <    MEAN2    <=    M2TARGET    THEN
DM2=((MEAN2-M2LOW)/(M2TARGET-M2LOW))**W1_MEAN2;
ELSE    IF    M2TARGET    <    MEAN2    <    M2HIGH    THEN
DM2=((M2HIGH-MEAN2)/(M2HIGH-M2TARGET))**W2_MEAN2;
ELSE IF M2HIGH <= MEAN2          THEN DM2=0;

        DM=(DM1*DM2)**(1/2);

        S1LOW=0.6;  S1HIGH=0.9;   W_SD1=&W_SD1;
        S2LOW=0.025; S2HIGH=0.0375; W_SD2=&W_SD2;

     IF SD1     <=S1LOW        THEN DS1=1;
 ELSE IF S1LOW  <=SD1< S1HIGH THEN DS1=((S1HIGH-SD1)/(S1HIGH-S1LOW))**W_SD1;
 ELSE IF S1HIGH <=SD1          THEN DS1=0;

     IF SD2     <=S2LOW        THEN DS2=1;
 ELSE IF S2LOW  <=SD2< S2HIGH THEN DS2=((S2HIGH-SD2)/(S2HIGH-S2LOW))**W_SD2;
 ELSE IF S2HIGH <=SD2          THEN DS2=0;

        DS=(DS1*DS2)**(1/2);
        W_MEAN=&W_MEAN;

        IF W_MEAN=1 THEN D=DM;
   ELSE IF W_MEAN=0 THEN D=DS;
   ELSE D=(DM**W_MEAN)*(DS**(1-W_MEAN));

DATA DD; SET D; IF D>0; RUN;
```

```
PROC SORT DATA=DD; BY DESCENDING D; RUN;
DATA OPT; SET DD; IF _N_=1;

TITLE 'Multiresponse Optimization Results';
DATA SUMMARY; SET OPT; W_SD=1-W_MEAN;
    LABEL W_MEAN='Weight for Optimization of Means'
          W_SD='Weight for Optimization of SDs'
          DM='Desirability of Means' DS='Desirability of SDs'
          D ='Overall Desirability of Means and SDs'
          X1='X1 Level' X2='X2 Level' X3='X3 Level';

PROC PRINT DATA=SUMMARY LABEL NOOBS; VAR W_MEAN W_SD D DM DS X1 X2 X3;
RUN;

DATA Y1MEAN; SET OPT;
    RESPONSE='Y1 Mean'; GOAL='Maximum'; LOW=M1LOW; HIGH=M1HIGH;
    WEIGHT1=W_MEAN1; D=DM1; VALUE=MEAN1;

DATA Y2MEAN; SET OPT;
    RESPONSE='Y2Mean'; GOAL='Target'; LOW=M2LOW; TARGET=M2TARGET; HIGH=M2HIGH;
    WEIGHT1=W1_MEAN2; WEIGHT2=W2_MEAN2; D=DM2; VALUE=MEAN2;

DATA Y1SD; SET OPT;
    RESPONSE='Y1 SD'; GOAL='Minimum'; LOW=S1LOW; HIGH=S1HIGH;
    WEIGHT1=W_SD1; D=DS1; VALUE=SD1;

DATA Y2SD; SET OPT;
    RESPONSE='Y2 SD'; GOAL='Minimum'; LOW=S2LOW; HIGH=S2HIGH;
    WEIGHT1=W_SD2; D=DS2; VALUE=SD2;

DATA MEANSD; SET Y1MEAN Y2MEAN Y1SD Y2SD;
  LABEL RESPONSE='Response Name' GOAL='Goal' LOW='Low' TARGET='Target' HIGH='High'
        WEIGHT1='Weight 1' WEIGHT2='Weight 2' D='Desirability' VALUE='Response Value';

PROC PRINT DATA=MEANSD NOOBS LABEL;
    VAR RESPONSE GOAL LOW TARGET HIGH WEIGHT1 WEIGHT2 D VALUE;
RUN;

%MEND;
```

위 SAS 코드의 데이터 부분에서는 X1, X2, X3 각각의 수준을 -2에서 2까지 0.1의 증분(increment)으로 증가시켜 가며 격자점들을 만든다. {-2, -1.9, ⋯, 1.9, 2}는 41개의 숫자들로 이루어지므로, 전체 격자점들의 개수는 $41 \times 41 \times 41 = 68,921$ 이 된다. 이 68,921개의 격자점들 중에서 D의 값을 최대화하는 X1, X2, X3의 수준값들을 찾는 것이다.

위 SAS 코드에서 데이터 다음 부분은 MULTIOPT라는 이름의 매크로로 되어 있으므로, 매크로 함수의 변수(argument) 값을 바꾸어가며 매크로를 수행시켜 여러 가지 방법으로 최적화를 시도할 수 있다.

먼저 w=0.5로 하고, 바람직성 함수에서의 멱 값들을 다 1로 하여 최적화를 수행하기 위하여 다음과 같이 매크로를 수행시켜 본다.

```
%MULTIOPT (W_MEAN=0.5, W_MEAN1=1, W1_MEAN2=1, W2_MEAN2=1, W_SD1=1, W_SD2=1);
```

이 매크로 수행의 출력은 다음과 같다.

Multiresponse Optimization Results

Weight for Optimization of Means	Weight for Optimization of SDs	Overall Desirability of Means and SDs	Desirability of Means	Desirability of SDs	X1 Level	X2 Level	X3 Level
0.5	0.5	0.62922	0.91850	0.43104	−0.4	−1.8	−0.3

Response Name	Goal	Low	Target	High	Weight 1	Weight 2	Desirability	Response Value
Y1 Mean	Maximum	8.000	.	12.0000	1	.	0.93369	11.7348
Y2 Mean	Target	0.700	0.75	0.8000	1	1	0.90356	0.7452
Y1 SD	Minimum	0.600	.	0.9000	1	.	0.34455	0.7966
Y2 SD	Minimum	0.025	.	0.0375	1	.	0.53925	0.0308

표준편차들에 관계없이 평균들만을 바람직성 함수에서의 멱들을 1로 하여 최적화하고 싶으면, 다음과 같이 매크로를 수행시킨다.

```
%MULTIOPT (W_MEAN=1, W_MEAN1=1, W1_MEAN2=1, W2_MEAN2=1, W_SD1=1, W_SD2=1);
```

이 매크로 수행의 출력은 다음과 같다. 최적점이 달라지는데, 표준편차들의 값들은 적절치 못함을 볼 수 있다.

Multiresponse Optimization Results

Weight for Optimization	Weight for Optimization	Overall Desirability of Means	Desirability	Desirability	X1	X2	X3

of Means	of SDs	and SDs	of Means	of SDs	Level	Level	Level
1	0	0.99962	0.99962	0	-1.6	-1.6	1.7

Response Name	Goal	Low	Target	High	Weight 1	Weight 2	Desirability	Response Value
Y1 Mean	Maximum	8.000	.	12.0000	1	.	1.00000	13.2284
Y2 Mean	Target	0.700	0.75	0.8000	1	1	0.99924	0.7500
Y1 SD	Minimum	0.600	.	0.9000	1	.	0.00000	1.1580
Y2 SD	Minimum	0.025	.	0.0375	1	.	0.00000	0.0783

평균들에 관계없이 표준편차들만을 바람직성 함수에서의 멱들을 1로 하여 최적화하고 싶으면, 다음과 같이 매크로를 수행시킨다.

```
%MULTIOPT (W_MEAN=0, W_MEAN1=1, W1_MEAN2=1, W2_MEAN2=1, W_SD1=1, W_SD2=1);
```

이 매크로 수행의 출력은 다음과 같다. 최적점이 달라지는데, Y2의 평균값은 받아들일 수 없는 값이 됨을 볼 수 있다.

Multiresponse Optimization Results

Weight for Optimization of Means	Weight for Optimization of SDs	Overall Desirability of Means and SDs	Desirability of Means	Desirability of SDs	X1 Level	X2 Level	X3 Level
0	1	0.47415	0	0.47415	-0.3	-2.0	-0.5

Response Name	Goal	Low	Target	High	Weight 1	Weight 2	Desirability	Response Value
Y1 Mean	Maximum	8.000	.	12.0000	1	.	0.90942	11.6377
Y2 Mean	Target	0.700	0.75	0.8000	1	1	0.00000	0.6865
Y1 SD	Minimum	0.600	.	0.9000	1	.	0.44647	0.7661
Y2 SD	Minimum	0.025	.	0.0375	1	.	0.50354	0.0312

평균들의 최적화를 더 중시하고, 각 바람직성 함수에서의 멱에도, 각 반응의 최적화의 상대적 중요성을 감안하여, 적절히 다른 값을 주어 평균과 표준편차들을 최적화하고자 한다. 이럴 경우, 다음과 같이 매크로를 수행시킬 수 있다.

```
%MULTIOPT (W_MEAN=0.8, W_MEAN1=2, W1_MEAN2=4, W2_MEAN2=4, W_SD1=2, W_SD2=1);
```

이 매크로 수행의 출력은 다음과 같다. 최적점이 달라지는데, 평균들의 최적성이 향상되었음을 볼 수 있다.

Multiresponse Optimization Results

Weight for Optimization of Means	Weight for Optimization of SDs	Overall Desirability of Means and SDs	Desirability of Means	Desirability of SDs	X1 Level	X2 Level	X3 Level
0.8	0.2	0.70071	0.97651	0.18578	-0.4	-1.8	-0.1

Response Name	Goal	Low	Target	High	Weight 1	Weight 2	Desirability	Response Value
Y1 Mean	Maximum	8.000	.	12.0000	2	.	0.96098	11.9212
Y2 Mean	Target	0.700	0.75	0.8000	4	4	0.99228	0.7501
Y1 SD	Minimum	0.600	.	0.9000	2	.	0.11872	0.7966
Y2 SD	Minimum	0.025	.	0.0375	1	.	0.29071	0.0339

평균들의 최적화를 더욱 더 중시하여 평균들의 최적화 가중치를 0.9로 하고, 평균들의 바람직성 함수들에서의 멱들에는 바로 전의 값들을 주고자 한다. 그런데, 각 표준편차는 받아들일 수 있는 범위 내의 값을 가지기만 하면 된다고 생각할 수 있다. 이런 상황에서는 표준편차들의 바람직성 함수에서의 멱들에 0의 값을 준다. 이 경우의 매크로 수행명령 코드는 다음과 같다.

```
%MULTIOPT (W_MEAN=0.9, W_MEAN1=2, W1_MEAN2=4, W2_MEAN2=4, W_SD1=0, W_SD2=0);
```

이 매크로 수행의 출력은 다음과 같다. 최적점이 달라지는데, 평균들의 최적성이 더욱 향상되어 Y1의 평균은 최적요건을 충족시키고 Y2의 평균은 최적요건을 거의 충족시키고 있음을 볼 수 있다.

Multiresponse Optimization Results

Overall

Weight for Optimization of Means	Weight for Optimization of SDs	Desirability of Means and SDs	Desirability of Means	Desirability of SDs	X1 Level	X2 Level	X3 Level
0.9	0.1	0.98624	0.98473	1	-0.5	-1.8	0.0

Response Name	Goal	Low	Target	High	Weight 1	Weight 2	Desirability	Response Value
Y1 Mean	Maximum	8.000	.	12.0000	2	.	1.00000	12.0980
Y2 Mean	Target	0.700	0.75	0.8000	4	4	0.96968	0.7504
Y1 SD	Minimum	0.600	.	0.9000	0	.	1.00000	0.8234
Y2 SD	Minimum	0.025	.	0.0375	0	.	1.00000	0.0342

연구자가 위 출력에 나타난 (x1, x2, x3)=(-0.5, -1.8, 0.0)을 최적점으로 선택하기로 하였다고 하자. 이제 바로 위 출력에 나타나 있는 가중치들을 가지고 종합 바람직성 함수(D)의 등고선 그림(contour plot)들을 그려 보도록 하자. 등고선 그림에서는 두 개의 요인들만 사용가능하므로, 처음 등고선 그림에서는 x3=0으로 하고 x1-x2 평면상에서 D의 등고선들을 그리도록 하자. 두 번째 등고선 그림에서는 x2=-1.8로 하고 x1-x3 평면상에서 D의 등고선들을 그리도록 하고, 세 번째 등고선 그림에서는 x1=-0.5로 하고 x2-x3 평면상에서 D의 등고선들을 그리도록 하자. 등고선들은 D=0.1, 0.2, 0.3, 0.4, 0.5, 0.6, 0.7, 0.8, 0.9, 0.98624(최대값)인 선들만 그리도록 한다. 이를 수행시키는 SAS 코드는 다음과 같다.

```
TITLE 'Contour Plot of Overall Desirability at X3=0';
PROC GCONTOUR DATA=D;
    PLOT X1*X2=D/ LEVELS=0.1 TO 0.9 BY 0.1 0.98624;
    WHERE ABS(X3)<10**(-10);
RUN;
TITLE 'Contour Plot of Overall Desirability at X2=-1.8';
PROC GCONTOUR DATA=D;
    PLOT X1*X3=D/ LEVELS=0.1 TO 0.9 BY 0.1 0.98624;
    WHERE ABS(X2-(-1.8))<10**(-10);
RUN;
TITLE 'Contour Plot of Overall Desirability at X1=-0.5';
PROC GCONTOUR DATA=D;
    PLOT X2*X3=D/ LEVELS=0.1 TO 0.9 BY 0.1 0.98624;
    WHERE ABS(X1-(-0.5))<10**(-10);
RUN;
```

앞의 SAS code의 출력은 다음과 같이 나타나 있다. 이 등고선 그림들을 검토한 결과, D의 값이 0.1 이상인 지역, 즉 연구자가 받아들일 수 있는 (acceptable) 추정된 반응값들을 생성하는 x1-x2-x3 공간이 전체 실험공간에서 차지하는 부분이 꽤 작다고

볼 수 있다.


Contour Plot of Overall Desirability at X3=0
X1
X2
D 0.1000 0.2000 0.3000 0.4000 0.5000
0.6000 0.7000 0.8000 0.9000 0.9862


Contour Plot of Overall Desirability at X1=-0.5
X2
X3
D 0.1000 0.2000 0.3000 0.4000 0.5000
0.6000 0.7000 0.8000 0.9000 0.9862


Contour Plot of Overall Desirability at X2=-1.8
X1
X3
D 0.1000 0.2000 0.3000 0.4000 0.5000
0.6000 0.7000 0.8000 0.9000 0.9862

3. 요약 및 결론

본고에서는 양적 품질특성치들이 둘 이상 있을 때의 다구찌의 파라미터 설계 문제를 반응표면적 방법으로 해결하는 방법을 실제 자료에 의한 예시와 더불어 구체적으로 제안하였다. 특히, 적절한 통계적 방법의 유효성을 인정하는 경우에도 그 통계적 방법을 실행하기 위한 프로그래밍 방법을 모를 때에는 사실상 그러한 통계적 방법의 실행이 불가능하다는 사실을 감안하여 예들에서 완벽한 SAS code들을 제공하였다. 이 SAS code들은 여러 분석의 수행방법을 알려주고 있는데, 이 SAS 코드들을 자세히 검토함으로써 독자들은 본고에서 제안하는 방법론을 더욱 확실하게 이해할 수 있을 것이다.

본고에서의 주요 연구 결과들은 (i) 2차 다항 모형에서의 적합결여를 극복하기 위한 3차 다항 모형의 모형 항 선택, (ii) 여러 품질특성치들의 기본모형들의 SAS PROC SYSLIN에 의한 FGLS 추정, (iii) 평균들과 표준편차들을 평균들의 최적화에 대한 가중치를 부여하여 동시에 최적화하는 종합 바람직성 함수의 제안으로 요약될 수 있다.

본고에서 제안하는 방법론을 실무에 적용할 때에는 추정된 최적점들에서 반드시 확인실험을 해 보아야 한다는 것을 명기(明記)한다. 필자들은 이 방법론이 산업현장에 적용되어 made in Korea 제품의 품질이 향상되는 데 기여하기를 바라고 있다.

참고문헌

[1] 박유성, 송석헌(1998), 「SAS/ETS를 활용한 경영경제자료분석」, 정일출판사.

[2] 이우선, 김영주(1997), "Desirability 함수 기법에 의한 다중반응표면의 최적화 연구," 「응용통계」, 제 12권, pp. 81-105.

[3] 임성수(1995), "실험계획을 통한 품질향상에의 반응표면적 접근방법: 다구찌 Parameter Design 방법에의 대안," 「응용통계」, 제 10권, pp. 29-41.

[4] Box, G.E.P., Hunter, W.G., and Hunter, J.S.(1978), *Statistics for Experimenters*, Wiley, New York.

[5] Derringer, D. and Suich, R.(1980), "Simultaneous Optimization of Several Response Variables," *Journal of Quality Technology*, Vol. 12, pp. 214-219.

[6] Myers, R.H. and Carter, W.H.(1973), "Response Sutface Techniques for Dual Response Systems," *Technometrics*, Vol. 15, pp. 301-317.

[7] Myers, R.H. and Montgomery, D.C.(1995), *Response Surface Methodology*, Wiley, New York.

[8] Myers, R.H., Khuri, A.I. and Vining, G.(1992), "Response Surface Alternatives to the Taguchi Robust Parameter Design Approach," *The American Statistician*, Vol. 46, pp. 131-139.

[9] Khuri, A.I. and Conlon, M.(1981), "Simultaneous Optimization of Multiple Response Represented by Polynomial Regression Functions," *Technometrics*, Vol. 23, pp. 363-375.

[10] Khuri, A.I. and Cornell, J.A.(1987), *Response Surfaces,* Marcel Dekker, New York.

[11] Vining, G.G, and Myers, R.H.(1990), "Combining Taguchi and Response Surface Philosophies: A Dual Response Approach," *Journal of Quality Technology*, Vol. 22, pp. 38-45.